

# PyAfipWs: Interfase COM para Factura Electrónica

## Manual de Uso

2008/2009 © Mariano Reingart – Versión 1.11– Mayo 2009

### Instalación:

Bajar y Ejecutar el autoinstalable [www.nsis.com.ar/soft/pyafipws/demo.exe](http://www.nsis.com.ar/soft/pyafipws/demo.exe)

- La interfase se instala automáticamente en C:\PYAFIPWS
- Instalación manual: `pyafipws.exe --register` o utilizar `regsvr32 pyafipws.dll`
- Desinstalación: `pyafipws.exe --unregister` o utilizar `regsvr32 /u pyafipws.dll`

Un ejemplo completo del uso de la interfase en Visual Basic se puede descargar desde [www.nsis.com.ar/soft/pyafipws/ejemplo.zip](http://www.nsis.com.ar/soft/pyafipws/ejemplo.zip)

Para más información, visitar <http://www.nsis.com.ar/public/wiki/PyAfipWs> o enviar mail a [pyafipws@nsis.com.ar](mailto:pyafipws@nsis.com.ar)

### Servicio Web de Autenticación y Autorización (WSAA)

Este servicio es necesario para acceder al resto de los servicios de negocios (ej. Factura Electrónica) ofrecidos por la AFIP.

Se debe generar un Ticket de Requerimiento de Acceso (TRA), firmarlo criptográficamente (generando un mensaje CMS) que es envía al servicio web para obtener un Ticket de Acceso (TA).

Esta interfase crea el mensaje en XML, lo firma criptográficamente y lo codifica en base-64 como lo requiere la AFIP, por lo que en general no es necesario realizar ninguno de estos pasos.

Para más información ver especificaciones técnicas del WSAA en el sitio de la AFIP: [https://wshomo.afip.gov.ar/wsfedocs/Especificacion\\_Tecnica\\_WSAA\\_1.2.0.pdf](https://wshomo.afip.gov.ar/wsfedocs/Especificacion_Tecnica_WSAA_1.2.0.pdf)

### Objeto:

- El objeto COM se crea invocando a `CreateObject("WSAA")`

### Métodos:

- `CreateTRA()`: devuelve un ticket de requerimiento de acceso (string con formato xml). Se puede revisar o modificar el xml devuelto (en general no es necesario)
- `SignTRA(tra, certificado, clave_privada)`: recibel el ticket de requerimiento de acceso (string con formato xml, creado previamente) y las ubicaciones del certificado y clave privada. Firma el ticket y devuelve el mensaje CMS (string con formato xml).
- `CallWSAA(cms, url_webservice)`: Recibe el mensaje CMS (string con formato xml) y la URL del webservice. Se conecta con el servidor remoto, ejecuta el método remoto `LoginCMS` y devuelve el Ticket de Acceso (TA, string con formato xml) generado por la AFIP. Si no se especifica url, se utiliza servidores de homologación.

### Atributos:

- `Token`: el el código de autorización generado por la AFIP, necesario para utilizar el resto de los servicios web.
- `Sign`: el la firma de autorización generado por la AFIP, necesario para utilizar el resto de los servicios web.
- `Version`: versión de la interfase (ej. "1.6")

## Ejemplo:

```
Dim WSAA As Object

' Crear objeto interface Web Service Autenticación y Autorización
Set WSAA = CreateObject("WSAA")

' Generar un Ticket de Requerimiento de Acceso (TRA)
tra = WSAA.CreateTRA()

' Especificar la ubicación de los archivos certificado y claves
Path = CurDir() + "\"
Certificado = "ghf.crt" ' certificado es el firmado por la afip
ClavePrivada = "ghf.key" ' clave privada usada para crear el cert.

' Generar el mensaje firmado (CMS)
cms = WSAA.SignTRA(tra, Path + Certificado, Path + ClavePrivada)

' Llamar al web service para autenticar:
ta = WSAA.CallWSAA(cms, url_web_service)

' Obtener Token y Sign de autorización
Token = WSAA.Token
Sign = WSAA.Sign
```

El Ticket de Acceso contiene el Token (código) y Sign (firma) son los elementos que se utilizan en los otros servicios web para lograr acceso a los mismos. Estos datos se pueden usar por un período de tiempo (5 horas, modificable), siendo necesario volver a invocar el WSAA solo al finalizar este período, para obtener nuevos Token y Sign.

## Errores:

El WSAA informa errores alfanuméricos (Err.Number – vbObjectError = 0), debiendo verificar la descripción (Err.Description). Entre ellos los más comunes son:

- `coe.notAuthorized, cms.cert.expired, cms.cert.invalid, cms.cert.untrusted`: El certificado no fue firmado por la AFIP, expiró, es inválido, no se confía en su emisor, o no coincide la fuente. Se debe revisar o emitir correctamente los certificados.
- `xml.generationTime.invalid, xml.expirationTime.expired, xml.expirationTime.invalid`: El tiempo de generación/expiración es inválido. Debe sincronizar relojes con el servidor de la AFIP.
- `wsn.unavailable, wsaa.unavailable`: servicio no disponible momentáneamente

## Servicio Web de Factura Electrónica (WSFE)

Este servicio es ofrecido por la AFIP para la facturación electrónica.

Para poder acceder, se debe obtener un Ticket de Acceso (TA) previamente (para utilizar el Token y Sign) y enviarlos a los métodos de facturación electrónica).

Esta interfase recibe los valores y crea los mensaje en XML, lo envía y analiza la respuesta XML, por lo que no es necesario modificar XML. Se recomienda pasar los valores en formato string, números separados por punto y fechas YYYYMMDD.

Para más información ver especificaciones técnicas del WSFE en el sitio de la AFIP:

<https://wswhomo.afip.gov.ar/wsfedocs/WSFE%20-%20Manual%20para%20el%20desarrollador%20-%2020070307.pdf>

## Objeto:

- El objeto COM se crea invocando a `CreateObject("WSFE")`

## Métodos:

- `Conectar(url)`: realiza la conexión a los servidores de la AFIP (primer paso esencial). Si no se especifica `ur`, se utiliza servidores de homologación.
- `Dummy()`: servicio de prueba para obtener el estado de los servidores de la AFIP.
- `UltNro()`: recuperar el último número de transacción (`id`). Devuelve el último valor de secuencia utilizado al pedir autorización. Uso opcional.
- `Aut(id, presta_serv, tipo_doc, nro_doc, tipo_cbte, punto_venta, cbt_desde, cbt_hasta, imp_total, imp_tot_conc, imp_netto, imppto_liq, imppto_liq_rni, imp_op_ex, fecha_cbte, fecha_venc_pago, [fecha_serv_desde], [fecha_serv_hasta])`: autoriza la emisión de factura electrónica, recibe el `id` de secuencia (recuperado por `UltNro`), `presta_serv` en 1 si se esta facturando servicios (en este caso es obligatorio el uso de la fecha de servicio desde y hasta), los datos de la factura a emitir, y devuelve el Código de Autorización Electrónico (CAE). Ver ejemplo para el detalle de los parámetros.
- `RecuperaLastCMP(punto_venta, tipo_comprobante)`: recibe el punto de venta y tipo de comprobante (strings/enteros), devuelve el último número de comprobante autorizado. Uso opcional.
- `RecuperaQty()` recupera cantidad máxima de registros de detalle (por el momento no es de mayor utilidad ya que por simplicidad se envía 1 registro de detalle por autorización). Uso opcional.

## Atributos:

- `Token`: es el código de autorización generado por la AFIP (WSAA)
- `Sign`: es la firma de autorización generado por la AFIP (WSAA)
- `Cuit`: es el número de CUIT del emisor de facturas, formato string sin guiones.
- `AppServerStatus, DbServerStatus, AuthServerStatus`: estados de los servidores de AFIP (string "OK" en caso de estar funcionales)
- `XmlRequest, XmlResponse`: requerimiento y respuesta XML (para depuración)
- `Respuesta, Motivo, RetProceso`: valores complementarios que retorna `Aut`
- `CAE, Vencimiento`: CAE y Fecha de vencimiento autorización retorna `Aut`
- `Version`: versión de la interfase (ej. "1.6")

## Errores:

La interfase procesa los errores WSFE y los informa con códigos numéricos (`Err.Number - vbObjectError = n° de error de AFIP`) y su correspondiente descripción (`Err.Description`).

Entre ellos los más comunes son:

- 1000: Usuario no autorizado a realizar esta operación. Revisar `Token` y `Sign` obtenidos del WSAA
- 1001: CUIT no autorizado. Revisar en el sitio de la AFIP si se asoció el CUIT del emisor al servicio de factura electrónica, con el correspondiente certificado.
- 1012, 1014, 1016, 1017, 1018, 1019: Inconsistencia en los datos enviados. Revisar formatos (fechas, números, etc., usar fechas YYYYMMDD y números separados por puntos, ambos pasarlos como string para evitar errores). Revisar tipo de datos (`tipo_cbte`, `tipo_doc`) que correspondan a las tablas usadas por la AFIP. Revisar que los datos sean válidos para la operación a realizar (ej. Factura A solo para `tipo_doc = 80` CUIT, Facturas B por lote hasta \$1000.-, etc.)

Ver especificaciones técnicas del WSFE para más información sobre el uso del servicio web, su formato, condiciones y errores.

## Ejemplo:

```
' Crear objeto interface Web Service de Factura Electrónica
Set WSFE = CreateObject("WSFE")
' Setear token y sing de autorización (pasos previos)
WSFE.Token = WSAA.Token
WSFE.Sign = WSAA.Sign

' CUIT del emisor (debe estar registrado en la AFIP)
WSFE.Cuit = "30710381400"

' Conectar al Servicio Web de Facturación (servidor de homologación)
ok = WSFE.Conectar("https://wshomo.afip.gov.ar/wsfe/service.asmx")

' Llamo a un servicio "tonto", para obtener el estado (opcional)
WSFE.Dummy
Debug.Print "appserver status", WSFE.AppServerStatus
Debug.Print "dbserver status", WSFE.DbServerStatus
Debug.Print "authserver status", WSFE.AuthServerStatus

' Recupera cantidad máxima de registros (opcional)
qty = WSFE.RecuperarQty()

' Recupera último número de secuencia ID
LastId = WSFE.UltNro()

' Recupero último número de comprobante (opcional)
tipo_cbte = 1: punto_vta = 1
LastCBTE = WSFE.RecuperaLastCMP(punto_vta, tipo_cbte)

' Establezco los valores de la factura o lote a autorizar:
Fecha = Format(Date, "yyyymmdd")
id = LastId + 1: presta_serv = 1
tipo_doc = 80: nro_doc = "2311111113"
cbt_desde = LastCBTE + 1: cbt_hasta = LastCBTE + 1
imp_total = "121.00": imp_tot_conc = "0.00": imp_neto = "100.00"
imppto_liq = "21.00": imppto_liq_rni = "0.00": imp_op_ex = "0.00"
fecha_cbte = Fecha: fecha_venc_pago = Fecha
' Fechas del período del servicio facturado (solo si presta_serv = 1)
fecha_serv_desde = Fecha: fecha_serv_hasta = Fecha

' Llamo al Webservice de Autorización para obtener el CAE
cae = WSFE.Aut(id, presta_serv, _
    tipo_doc, nro_doc, tipo_cbte, punto_vta, _
    cbt_desde, cbt_hasta, imp_total, imp_tot_conc, imp_neto, _
    imppto_liq, imppto_liq_rni, imp_op_ex, fecha_cbte, fecha_venc_pago,
    fecha_serv_desde, fecha_serv_hasta) ' solo pasar si presta_serv=0

MsgBox "CAE: " & cae & " Vencimiento: " & WSFE.Vencimiento
Exit Sub
ManejoError:
    Si hubo error:
        Debug.Print Err.Description ' descripción error afip
        Debug.Print Err.Number - vbObjectError ' código error afip
        MsgBox Err.Description, vbCritical + vbRetryCancel, "Error:" &
Err.Number - vbObjectError & " en " & Err.Source
```

## Descripción del Método Aut (obtención de CAE):

Se debe llamar al método `Aut` con los siguiente parámetros:

- `id`: Número de identificación secuencial (debe almacenarse en el sistema local) No es obligatorio llamar a `UltNro`, puede utilizarse un dato local secuencial.
- `presta_serv`: 0 o 1 para indicar si es una factura de servicios
- `tipo_doc`, `nro_doc`: Tipo (80 CUIT, 96 DNI, etc.) y número de Documento
- `tipo_cbte`: Tipo de comprobante (1 Factura A, 6 Factura B, etc.)
- `punto_vta`: N° de punto de venta (debe estar autorizado)
- `cbt_desde`, `cbt_hasta`: N° de comprobante (si es un solo comprobante, repetir N°)
- `imp_total`: Importe total de la factura
- `imp_tot_conc`: Importe total de conceptos no gravados por el IVA
- `imp_netto`: Importe neto (gravado por el IVA) de la factura
- `impto_liq`: Importe del IVA liquidado (incluyendo percepciones de IVA)
- `impto_liq_rni`: Importe IVA RNI (no se utiliza mas, dejar 0,00)
- `imp_op_ex`: Importe de operaciones exentas
- `fecha_cbte`: Fecha del comprobante (no puede ser mayor o menor a 5 días)
- `fecha_venc_pago`: Fecha límite de vencimiento para el pago de la factura
- `fecha_serv_desde`, `fecha_serv_hasta`: Fechas del período de servicios prestado

Ver todos los tipos de comprobante: <https://wsw homo.afip.gov.ar/wsfedocs/CodCpbte.txt>

Ver todos los tipos de documentos: <https://wsw homo.afip.gov.ar/wsfedocs/CodDocID.txt>

Si no hubiere inconvenientes, la llamada debe devolver el CAE y se establece el atributo `WSFE.Resultado = "A"` (Aceptado) y `WSFE.Vencimiento`. Sino, devuelve:

- `cae = ""`: La interfase no pudo procesar la respuesta del Webservice, o bien la conexión a internet esta caída, las direcciones de los servidores son incorrectas, o los servidores de la AFIP no funcionan o no devuelven la respuesta esperada.
- `cae = "NULL"`: El servicio web rechazo los datos de la factura a generar. Se establece el atributo `WSFE.Resultado = "R"` (Rechazado) y `WSFE.Motivo` con los diversos motivos de rechazo proporcionados por el webservice:
  - `WSFE.Motivo = "00"`: "NULL": no hay error (solo como referencia)
  - `WSFE.Motivo = "01"`: CUIT informada no es R.I.
  - `WSFE.Motivo = "02"`: CUIT no autorizada a facturar electrónicamente
  - `WSFE.Motivo = "03"`: CUIT registra inconvenientes con domicilio fiscal
  - `WSFE.Motivo = "04"`: Punto de venta no se encuentra declarado
  - `WSFE.Motivo = "05"`: Fecha del comprobante incorrecta
  - `WSFE.Motivo = "06"`: CUIT no puede emitir comprobantes clase A
  - `WSFE.Motivo = "07"`: para comprobantes clase A debe indicar CUIT
  - `WSFE.Motivo = "08"`: CUIT informada es inválida
  - `WSFE.Motivo = "09"`: CUIT informada no existe en el padrón
  - `WSFE.Motivo = "10"`: CUIT informada no corresponde a un R.I.
  - `WSFE.Motivo = "11"`: el n° de comprobante no es correlativo
  - `WSFE.Motivo = "12"`: el rango informado se encuentra autorizado

Aún devolviendo el CAE y con el resultado Aceptado, puede haber un motivo (ej 01), que funcionan como advertencia para revisar la base de datos y corregir posibles irregularidades.

**Advertencia:** En el caso de que la conexión se interrumpa luego de que la AFIP registre la factura y antes de que la información del CAE llegue al equipo, la única alternativa es repetir la llamada a `Aut` con los mismos datos (incluido el ID original) para obtener el CAE generado anteriormente. En ese caso, se `WSFE.Reproceso = "S"` para indicar que el CAE se ha recuperado. Por esto, es importante guardar el ID o utilizar un dato local.

Ver <https://wsw homo.afip.gov.ar/wsfedocs/WSFE-GuiaAdicionalParaElProgramador.pdf>

## Servicio Web de Bono Fiscal Electrónico (WSBFE)

EL WSBFE (Bonos Fiscales Electrónicos) es un Servicio Web de la AFIP para Facturas Electrónicas de Bienes de Capital, correspondiente al Artículo 3 de la Resolución General 2557/2009, Este servicio es ofrecidos por la AFIP para la facturación electrónica.

Para poder acceder, se debe obtener un Ticket de Acceso (TA) previamente (para utilizar el Token y Sign y enviarlos a los métodos de facturación electrónica).

Esta interfase recibe los valores y crea los mensaje en XML, lo envía y analiza la respuesta XML, por lo que no es necesario modificar XML. Se recomienda pasar los valores en formato string, números separados por punto y fechas YYYYMMDD.

La operatoria es bastante similar al método de autorización del WSFE (con un ID secuencial, observaciones, reproceso, etc.), teniendo en cuenta esta mayor complejidad por tener que informar el detalle de cada item.

A su vez, el WSBFE devuelve mensajes de eventos (mantenimiento programado, advertencias, etc.), los que deben ser capturados e informados al usuario.

A diferencia del WSFE, este nuevo servicio funciona con tablas dinámicas de parámetros para los códigos de comprobante, moneda, alícuotas de iva, producto según NCM, zonas, unidades de medida. Estas tablas pueden sufrir modificaciones realizadas por la AFIP, con altas y bajas lógicas, por lo que tienen una fecha de vigencia (desde, hasta) y se proveen métodos para consultarlas por el mismo servicio web (a diferencia del WSFE, que las tablas eran documentadas estáticamente en el sitio web).

Para más información ver especificaciones técnicas del WSBFE en el sitio de la AFIP: <http://wswhomo.afip.gov.ar/fiscaldocs/WSBFE/WSBFE-ManualParaElDesarrollador.pdf>

La programación de la interfase WSBFE es similar a WSFE, se recomienda revisarlo.

### Objeto:

- El objeto COM se crea invocando a `CreateObject("WSBFE")`

### Métodos:

- `Conectar(url)`: realiza la conexión a los servidores de la AFIP (primer paso esencial). Si no se especifica url, se utiliza servidores de homologación.
- `Dummy()`: servicio de prueba para obtener el estado de los servidores de la AFIP.
- `CrearFactura(tipo_doc, nro_doc, zona, tipo_cbte, punto_vta, cbte_nro, fecha_cbte, imp_total, imp_netto, imp_to_liq, imp_tot_conc, imp_to_liq_ini, imp_op_ex, imp_perc, imp_iibb, imp_perc_mun, imp_internos, imp_moneda_id, Imp_moneda_ctz)`: crea internamente una factura para luego poder autorizarla, recibe los datos de la factura a emitir. Ver ejemplo para el detalle de los parámetros.
- `AgregarItem(ncm, sec, ds, qty, umed, precio, bonif, iva_id, imp_total)`: agrega internamente un item (línea de factura) a una factura para luego poder autorizarla, recibe los datos del item a factura a emitir. Ver ejemplo para el detalle de los parámetros.
- `Authorize(id)`: autoriza la emisión de factura electrónica para bono fiscal electrónico, recibe el id de secuencia (recuperado por `WSFE.UltNro`), y devuelve el Código de Autorización Electrónico (CAE). Ver ejemplo para el detalle de los parámetros.

## Atributos:

- Token: es el código de autorización generado por la AFIP (WSAA)
- Sign: es la firma de autorización generado por la AFIP (WSAA)
- Cuit: es el número de CUIT del emisor de facturas, formato string sin guiones.
- AppServerStatus, DbServerStatus, AuthServerStatus: estados de los servidores de AFIP (string "OK" en caso de estar funcionales)
- XmlRequest, XmlResponse: requerimiento y respuesta XML (para depuración)
- Respuesta, Obs, Reproceso: valores complementarios que retorna Aut
- CAE, Vencimiento: CAE y Fecha de vencimiento autorización retorna Aut
- Version: versión de la interfase (ej. "1.11")

## Errores:

La interfase procesa los errores WSFE y los informa con códigos numéricos (Err.Number – vbObjectError = nº de error de AFIP) y su correspondiente descripción (Err.Description).

Entre ellos los más comunes son:

- 1000: Usuario no autorizado a realizar esta operación. Revisar Token y Sign obtenido del WSAA
- 1001: CUIT no autorizado. Revisar en el sitio de la AFIP si se asoció el CUIT del emisor al servicio de factura electrónica, con el correspondiente certificado.
- 1012, 1014, 1016, 1017, 1018, 1019: Inconsistencia en los datos enviados. Revisar formatos (fechas, números, etc., usar fechas YYMMDD y números separados por puntos, ambos pasarlos como string para evitar errores). Revisar tipo de datos (tipo\_cbte, tipo\_doc) que correspondan a las tablas usadas por la AFIP. Revisar que los datos sean válidos para la operación.

Ver especificaciones técnicas del WSBFE para más información sobre el uso del servicio web, su formato, condiciones y errores.

## Ejemplo:

```
Dim WSAA As Object, WSBFE As Object

' Crear objeto interface Web Service Autenticación y Autorización
Set WSAA = CreateObject("WSAA")

' Generar un Ticket de Requerimiento de Acceso (TRA) para WSBFE
tra = WSAA.CreateTRA("wsbfe")
' Especificar la ubicación de los archivos certificado y clave privada
Path = CurDir() + "\"
' Certificado: certificado es el firmado por la AFIP
' ClavePrivada: la clave privada usada para crear el certificado
Certificado = "reingart.crt" ' certificado de prueba
ClavePrivada = "reingart.key" ' clave privada de prueba
' Generar el mensaje firmado (CMS)
cms = WSAA.SignTRA(tra, Path + Certificado, Path + ClavePrivada)

' Llamar al web service para autenticar:
ta = WSAA.CallWSAA(cms,
"http://wsaahomo.afip.gov.ar/ws/services/LoginCms") ' Homologación

' Una vez obtenido, se puede usar el mismo token y sign por 24 horas

' Crear objeto interface Web Service de Factura Electrónica
Set WSBFE = CreateObject("WSBFE")
' Setear token y sign de autorización (pasos previos)
WSBFE.Token = WSAA.Token
WSBFE.Sign = WSAA.Sign

' CUIT del emisor (debe estar registrado en la AFIP)
WSBFE.Cuit = "20267565393"
```

```
' Conectar al Servicio Web de Facturación
ok = WSBFE.Conectar("http://wswhomo.afip.gov.ar/wsbfe/service.asmx") '
homologación
```

```
' Establezco los valores de la factura a autorizar:
```

```
fecha = "20090530"
```

```
tipo_doc = 80: nro_doc = "23111111113"
```

```
zona = 0 ' Ver tabla de zonas
```

```
tipo_cbte = 1 ' Ver tabla de tipos de comprobante
```

```
punto_vta = 2: cbte_nro = 16
```

```
fecha_cbte = fecha
```

```
imp_total = "121.00": imp_tot_conc = "0.00": imp_netto = "100.00"
```

```
imppto_liq = "21.00": imppto_liq_rni = "0.00": imp_op_ex = "0.00"
```

```
imp_perc = "0.00": imp_iibb = "0.00": imp_perc_mun = "0.00"
```

```
imp_internos = "0.00"
```

```
imp_moneda_id = "10" ' Ver tabla de tipos de moneda
```

```
Imp_moneda_ctz = "1.0000" ' cotización de la moneda
```

```
' Creo una factura (internamente, no se llama al Webservice):
```

```
ok = WSBFE.CrearFactura(tipo_doc, nro_doc, _
                        zona, tipo_cbte, punto_vta, cbte_nro, fecha_cbte, _
                        imp_total, imp_netto, imppto_liq, _
                        imp_tot_conc, imppto_liq_rni, imp_op_ex, _
                        imp_perc, imp_iibb, imp_perc_mun, imp_internos, _
                        imp_moneda_id, Imp_moneda_ctz)
```

```
' Agrego un item:
```

```
ncm = "7308.10.00" ' Ver tabla de códigos habilitados del NCM
```

```
sec = "" ' Código de la Secretaría (no usado por el momento)
```

```
ds = "Prueba" ' Descripción completa del artículo (hasta 4000 car.)
```

```
umed = 1 ' kg, Ver tabla de unidades de medida
```

```
qty = "1.0" ' cantidad
```

```
precio = "105.0" ' precio neto (Fac.A), precio final (Fac..uras B)
```

```
bonif = "5.00" ' descuentos (en positivo)
```

```
iva_id = 5 ' 21%, ver tabla de cuota de iva
```

```
imp_total = "121.00" ' importe total final del artículo
```

```
' lo agrego a la factura (internamente, no se llama al Webservice):
```

```
ok = WSBFE.AgregarItem(ncm, sec, ds, qty, umed, precio, bonif, iva_id,
imp_total)
```

```
' Llamo al Webservice de Autorización para obtener el CAE
```

```
cae = WSBFE.Authorize(id)
```

```
If cae = "" Or WSBFE.Resultado <> "A" Then
```

```
MsgBox "No se asignó CAE (Rechazado). Observación (motivos): " &
WSBFE.Obs, vbInformation + vbOKOnly
```

```
ElseIf WSBFE.Obs <> "" And WSBFE.Obs <> "00" Then
```

```
MsgBox "Se asignó CAE pero con advertencias. Observación
(motivos): " & WSBFE.Obs, vbInformation + vbOKOnly
End If
```

```
MsgBox "Resultado:" & WSBFE.Resultado & " CAE: " & cae & " Reproceso:
" & WSBFE.Reproceso & " Obs: " & WSBFE.Obs, vbInformation + vbOKOnly
```

```
' Muestro los eventos (mantenimiento programados y otros mensajes de
la AFIP)
```

```
For Each evento In WSBFE.Eventos
```

```
    If evento <> "0: " Then
```

```
        MsgBox "Evento: " & evento, vbInformation
```

```
    End If
```

```
Next
```

## Descripción de la obtención de CAE:

Como primer paso, se debe crear una factura (utilizada internamente por la interfase para contener los valores a autorizar), llamando al método `CrearFactura` con los siguiente parámetros:

- `tipo_doc, nro_doc`: Tipo (80 CUIT, 96 DNI, etc.) y número de Documento
- `tipo_cbte`: Tipo de comprobante (según tabla de parámetros)
- `punto_vta`: N° de punto de venta (debe estar autorizado para WSBFE)
- `cbt_nro`: N° de comprobante
- `fecha_cbte`: Fecha del comprobante (no puede ser mayor o menor a 10 días)
- `zona_id`: Zona (según tabla de parámetros) - Por el momento 0
- `imp_total`: Importe total de la factura
- `imp_tot_conc`: Importe total de conceptos no gravados por el IVA
- `imp_netto`: Importe neto (gravado por el IVA) de la factura
- `impto_liq`: Importe del IVA liquidado (incluyendo percepciones de IVA)
- `impto_liq_rni`: Importe IVA RNI (no se utiliza mas, dejar 0.00)
- `imp_op_ex`: Importe de operaciones exentas
- `imp_moneda_id`: Moneda de la factura (según tabla de parámetros)
- `imp_moneda_ctz`: Cotización de la moneda de la factura
- `imp_perc`: Importe de las precepciones
- `imp_iibb`: Importe de las percepciones de ingresos brutos
- `imp_perc_mun`: Importe de las percepciones municipales
- `imp_internos`: Importe de los impuestos internos

Luego, por cada artículo vendido (ítem), se debe llamar al método `AgregarItem` con los siguientes parámetros:

- `ncm`: código habilitado según el Nomenclador Común del Mercosur (NCM)
- `sec`: código de la secretaría (no utilizado por el momento, pasar '')
- `ds`: Descripción completa
- `precio`: Precio Neto Unitario (facturas A) o precio final (facturas B)
- `qty`: Cantidad
- `umed`: Unidad de medida (según tabla de parámetros)
- `bonif`: Bonificación
- `iva_id`: Alícuota de IVA (según tabla de parámetros)
- `imp_total`: Importe total (incluyendo descuentos e IVA)

Como último paso, se debe llamar al método `Authorize` con los siguiente parámetros:

- `id`: Numero de identificación secuencial (debe almacenarse en el sistema local) No es obligatorio llamar a `WSFE.UltNro`, puede utilizarse un dato local secuencial.

Si no hubiere inconvenientes, la llamada debe devolver el CAE y se establece el atributo `WSBFE.Resultado = "A"` (Aceptado) y `WSBFE.Vencimiento`. Sino, devuelve:

- `cae = ""`: La interfase no pudo procesar la respuesta del WebService, o bien la conexión a internet esta caída, las direcciones de los servidores son incorrectas, o el servicio web rechazo los datos de la factura a generar. Se establece el atributo `WSBFE.Resultado = "R"` (Rechazado) y `WSBFE.Obs` con los diversos motivos de rechazo proporcionados por el webservice (ver `WSFE.Motivo`):

Aún devolviendo el CAE y con el resultado Aceptado, puede haber un motivo (ej 01), que funcionan como advertencia para revisar la base de datos y corregir posibles irregularidades.

**Advertencia:** Revisar las consideraciones de reproceso al igual que con `WSFE`.

## Interfase por archivos de texto símil SIAP - RECE

Para lenguajes donde no es posible utilizar objetos COM, como en algunas versiones de COBOL, se desarrolló una interfase de texto para poder utilizar los Web Services de la AFIP, que funciona como un programa independiente, manteniendo las ventajas y características presentadas anteriormente.

### Configuración:

Editar el archivo RECE.INI en la carpeta de la interfase (C:\PYAFIPWS):

- CERT: ubicación del archivo certificado (ver WSAA)
- PRIVATEKEY: ubicación del archivo de la clave privada (ver WSAA)
- CUIT: CUIT del emisor
- ENTRADA: ubicación del archivo de texto de entrada
- SALIDA: ubicación del archivo de texto de salida
- URL: dirección de los servicios web de producción (ver WSAA y WSFE)

Ejemplo:

```
[WSAA]
CERT=C:\SISTEMA\empresa.crt
PRIVATEKEY= C:\SISTEMA\empresa.key
URL=https://wsaa.afip.gov.ar/ws/services/LoginCms
[WSFE]
CUIT=30000000000
ENTRADA= C:\SISTEMA\entrada.txt
SALIDA= C:\SISTEMA\salida.txt
URL=https://wsw.afip.gov.ar/wsfe/service.asmx
```

### Formato de archivos de entrada y salida:

El formato utilizado es similar a RECE pero extendido con algunos campos necesarios para utilizar los Servicios Web. A continuación se detalla nombre del campo, posición inicial, longitud y tipo:

- fecha\_cbte: (2,8,A)
- tipo\_cbte: (10,2,N)
- punto\_vta: (13,4,N)
- cbt\_desde: (17,8,N)
- cbt\_hasta: (25,8,N)
- tipo\_doc: (36,2,N)
- nro\_doc: (38,11,N)
- imp\_total: (79,15,I)
- imp\_tot\_conc: (94,15,I)
- imp\_netto: (109,15,I)
- impto\_liq: (124,15,I)
- impto\_liq\_rti: (139,15,I)
- imp\_cp\_ex: (154,15,I)
- cae: (261,14,N)
- fecha\_vto: (275,8,A)
- resultado: (291,1,A)
- motivo: (292,2,A)
- reproceso: (294,1,A)
- fecha\_venc\_pago: (295,8,A)
- presta\_serv: (303,1,N)
- fecha\_serv\_desde: (304,8,A)
- fecha\_serv\_hasta: (312,8,A)
- id: (320,15,N)

Tipos de campo:

- A: Alfanumerico
- N: Numerico
- I: Importes con 2 decimales (sin incluir punto)

Los espacios no definidos se ignoran (completar con un espacio en blanco)

Para más información sobre los campos, ver parámetros y valores que retorna el método `Aut` en el presente manual

Como formato de entrada es posible utilizar el mismo archivo generado para el aplicativo SIAP/RECE, pero no se podrán facturar servicios ya que los campos `presta_serv` y `fecha_serv_desde` y `fecha_serv_hasta` no están contemplados en dicho formato.

Si se informa el campo `ID`, la factura se genera con dicho número de secuencia. De lo contrario, para evitar problemas y poder reprocesar el CAE, se genera con un número de secuencia del formato `TPPPPNNNNNNNN` (tipo de comprobante, punto de venta y número de comprobante).

### Forma de uso:

Llamar al ejecutable `RECE.EXE` en la carpeta de la interfase (`C:\PYAFIPWS`)

En caso de ejecución correcta, informara por pantalla los `ID` y `CAE` obtenidos y el código de retorno es 0:

```
ID: 1000200000063 CAE: 59033567692234 Motivo: 00 Reproceso: N
ID: 1000200000064 CAE: 59033693441244 Motivo: 00 Reproceso: N
ID: 1000200000065 CAE: 59033880026543 Motivo: 00 Reproceso: N
```

El `CAE` obtenido, fecha de vencimiento y demás valores devueltos por `WSFE` (resultado, reproceso, motivo de rechazo u observación) son escritos en el archivo de salida, con la misma información de la factura que en el archivo de entrada.

En el caso de error, informa por pantalla el motivo y el código de retorno es distinto de 0:

```
ns1:coe.notAuthorized Computador no autorizado a acceder los
servicio de AFIP

1000 FERecuperaLastCMPRequest@SRV_RA: Usuario no autorizado a
realizar esta operacion. ValidacionDeToken: No apareció CUIT en
lista de relaciones: 100000000000
```

### Parámetros:

- `RECE.EXE /prueba`: Teniendo los certificados instalados, se puede realizar una prueba donde la interface generará un archivo de entrada para las tres próximas facturas, obteniendo los últimos números de transacción y comprobante.
- `RECE.EXE /ult`: Solicita Tipo de comprobante y Punto de Venta y devuelve el último número de comprobante registrado:

```
Consultar ultimo numero:
Tipo de comprobante: 1
Punto de venta: 2
Ultimo numero: 62
```

### Entorno:

Algunas versiones de `COBOL` cambian el directorio u otras variables del entorno (codificación, zona horaria, etc.). En este caso se recomienda armar un archivo `RECE.BAT`:

```
@ECHO OFF
REM limpiar la zona horaria y cambiar al directorio de la interfase
SET TZ=
CD C:\PYAFIPWS
RECE.EXE
```